*
* anim_tsurf.gs
*
* A simple animation of the diurnal cycle of surface temperature means
* during one season.  Useful for a quick check that the data has been
* read correctly; a warm region (local afternoon) should be seen to
* propagate from east to west across the map.
*
’open /home/ares/md1/esa/mcd/data/viks04me’
’set t 1 12’
’set lat -90 90’
’set lon -180 180’
’set mpdraw off’
’set grads off’
’set gxout contour’
’display tsurf’

*
* map_ps.gs
*
* Plots maps of the diurnal mean surface pressure and its (non-diurnal)
* standard deviation. GrADS should be run in portrait mode for this script.
*
’open /home/ares/md1/esa/mcd/data/viks04me’
’open /home/ares/md1/esa/mcd/data/viks04sd’
’set dfile 1’
’set vpage 0.0 8.5 5.5 11.0’
’set lat -90 90’
’set lon -180 180’
’define Pave = ave(ps,t=1,t=12)’
’set mpdraw off’
’set grads off’
’set gxout shaded’
’set cint 50’
’display Pave’
’set gxout contour’
’set ccolor 0’
’set cint 50’
’display Pave’
’draw xlab Diurnal Mean Surface Pressure (Pa)’
’draw title MCD Viking dust, Ls = 90-120.’
’set dfile 2’
’set vpage 0.0 8.5 0.0 5.5’
’set lat -90 90’
’set lon -180 180’
’set mpdraw off’
’set grads off’
’set gxout shaded’
’set cint 5’
’display sdps’
’set gxout contour’
’set ccolor 0’
’set cint 5’
’display sdps’
’draw xlab Diurnal Mean Std. Dev. Surface Pressure (Pa)’

*
* map_windt.gs
*
* Plots maps of the diurnal mean near-surface wind and surface temperature.
* GrADS should be run in portrait mode for this script.
*
’open /home/ares/md1/esa/mcd/data/viks04me’
’set vpage 0.0 8.5 5.35 10.45’
’set lat -90 90’
’set lon -180 180’
’set z 1’
’set mpdraw off’
’set grads off’
’define Uave = ave(u,t=1,t=12)’
’define Vave = ave(v,t=1,t=12)’
’set gxout vector’
’set arrlab off’
’set arrscl 0.4 20’
* Only display a vector at every other gridpoint in each direction to
* reduce clutter.
’display skip(Uave,2,2); Vave’
’draw xlab Diurnal Mean Low Level Wind (1mm = 2m/s)’
’draw title MCD Viking dust, Ls = 90-120.’
’set vpage 0.0 8.5 0.2 5.3’
’set lat -90 90’
’set lon -180 180’
’set mpdraw off’
’set grads off’
’define STave = ave(tsurf,t=1,t=12)’
’set gxout shaded’
’set cint 10’
’display STave’
’set gxout contour’
’set ccolor 0’
’set cint 10’
’display STave’
’draw xlab Diurnal Mean Surface Temperature (K)’

*
* profile.gs
*
* This script gives the mean, and confidence limits at a chosen number of
* standard deviations, for surface variables and also plots the vertical
* profile of three dimensional variables from the Mars Climate Database.
* It is important to note that the standard deviation merely represents the
* variability between profiles at that location and the same local solar
* time of day during that season.  It does not include any diurnal
* variability (this is respresented in the twelve means contained in the
* mean data file for different times of day) nor any small-scale
* variability nor any information about correlation and scales of
* variability within or between individual profiles (the last two points
* are addressed by the main database software).
*
* The longitude, latitude and local solar time are supplied as arguments.
* The longitude should be given in degrees east and the time in hours
* (00..24).  This script must be edited to select a dust scenario and
* season and to change the title accordingly.
*
* GrADS should be run in portrait mode.
*
* The script can be run from the GrADS command line, perhaps repeatedly, e.g.
* ga> run profile 000 00 00
* ga> c
* ga> run profile 000 00 12
* ....
*
* or in batch mode from the unix prompt, e.g.
* % grads -pbc ’run profile 000 00 00’
*
* A metafile, named using the plot parameters is written,
* e.g. profile_000_00_00.gx, which can then be viewed with gxtran
* or converted to postscript for printing with gxps or gxeps.
*
function main(args)
*
* Set longitude (-180..180 degrees east), latitude (-90..90 degrees north)
* and local solar time (00..24 hours) for profiles
plon = subwrd(args,1)
if (plon > 180); plon = plon - 360; endif
if (plon < -180); plon = plon + 360; endif
plat = subwrd(args,2)
ptime = subwrd(args,3)
if (ptime = ’’)
   say ’Three arguments are required for profile.gs’
   say ’run profile <longitude> <latitude> <time>’
   say ’longitude = -180..180 (degrees east)’
   say ’latitude = -90..90 (degrees north)’
   say ’local solar time = 00..24 (hours)’
endif
*
* Choose data files for means and deviations according to dust scenario and
* season required and modify the main title accordingly
mefile = ’/home/ares/md1/esa/mcd/data/viks04me.ctl’
sdfile = ’/home/ares/md1/esa/mcd/data/viks04sd.ctl’
title1 = ’MCD Viking dust, Ls = 90-120.’
*
* Choose the number of standard deviations for the confidence limits
nsds = 1
*
* Add profile location and time to title
title2 = ’Profile location:  lon = ’plon’E,  ’
title2 = title2%’lat = ’plat’N,  ’
title2 = title2%’local time = ’ptime’hrs.’
*
* If no data files are already open, open them,
* or if some are open check that they are what you expect them to be
’query files’
if (sublin(result, 1) = ’No files open’)
   ’open ’mefile
   ’open ’sdfile
else
   file1 = sublin(result, 2)
   file1 = subwrd(file1, 2)

   file2 = sublin(result, 5)
   file2 = subwrd(file2, 2)
   if ((file1!=mefile)|(file2!=sdfile))
      say ’The files open in grads are not consistent with profile.gs’
      say ’It is advisable to quit this session of grads and start again’
      return
   endif
endif
*
* Open metafile for writing output
’enable print profile_’plon’_’plat’_’ptime’.gx’
*
* Convert local time into universal time at lon=0
ptime = ptime - plon/15.0
if (ptime >= 24.0)
   ptime = ptime - 24.0
else
   if (ptime < 0.0); ptime = ptime + 24.0; endif
endif
*
* Choose the closest grid and time points present in the database.
* Note that there can be a problem with the current version of GrADS if a
* point with a latitude of exactly +/-90 degrees is chosen. In the case
* where data from the pole is chosen the latitude is shifted to 89.99 deg.
’set dfile 1’
’set lon ’plon
if (plat = 90); plat = 89.99; endif
if (plat = -90); plat = -89.99; endif
’set lat ’plat
ptime = (ptime + 1.0)/2.0
’set t ’ptime
*
* Title information
’set vpage 0.0 8.0 8.5 11.0’
’draw string 1.0 2.0 ’title1
’draw string 1.0 1.75 ’title2
subtitle = ’Mean data with confidence limits at ’nsds’ standard deviation’
if (nsds = 1)
   subtitle = subtitle%’.’
else
   subtitle = subtitle%’s.’
endif
’draw string 1.0 1.5 ’subtitle
*
* --- Surface data ---
’set z 1’
*
* surface pressure
’display ps’
sme = subwrd(result,4)
’display sdps.2(t=1)’
ssd = subwrd(result,4)
slo = max(sme - nsds*ssd, 0)
shi = sme + nsds*ssd
slo = round(slo, 0)
sme = round(sme, 0)
shi = round(shi, 0)
’draw string 1.0 1.0 Surface Pressure (Pa): ’slo’, ’sme’, ’shi
*
* surface temperature
’display tsurf’
sme = subwrd(result,4)
’display sdtsurf.2(t=1)’
ssd = subwrd(result,4)
slo = max(sme - nsds*ssd, 0)
shi = sme + nsds*ssd
slo = round(slo, 1)
sme = round(sme, 1)
shi = round(shi, 1)
’draw string 1.0 0.75 Surface Temperature (K): ’slo’, ’sme’, ’shi
*
* emissivity
’display emis’
sme = subwrd(result,4)
’display sdemis.2(t=1)’
ssd = subwrd(result,4)
slo = max(sme - nsds*ssd, 0.00)
shi = min(sme + nsds*ssd, 1.00)
slo = round(slo, 2)
sme = round(sme, 2)
shi = round(shi, 2)
’draw string 1.0 0.5 Emissivity: ’slo’, ’sme’, ’shi
*
* CO2 ice
’display co2ice’
sme = subwrd(result,4)
’display sdco2ice.2(t=1)’
ssd = subwrd(result,4)
slo = max(sme - nsds*ssd, 0)
shi = sme + nsds*ssd
slo = round(slo, 0)
sme = round(sme, 0)
shi = round(shi, 0)
’draw string 1.0 0.25 CO2 Ice (kg/m^2): ’slo’, ’sme’, ’shi
*
* --- Vertical profiles ---
’set z 1 25’
’set zlog on’
*
* temperature
’set vpage 0.0 4.0 4.5 8.5’
’set gxout line’
’set ylevs 1 0.5 0.2 0.1 0.05 0.02 0.01 0.005 0.002 0.001 0.0005 0.0002 0.0001’
’set cmark 0’
’set line 1’
’set grads off’
’display t’
’set cmark 0’
’set cstyle 5’
’set ccolor 1’
’set grads off’
’display t + ’nsds’*sdt.2(t=1)’
’set cmark 0’
’set cstyle 5’
’set ccolor 1’
’set grads off’
’display t - ’nsds’*sdt.2(t=1)’
’draw xlab Temperature (K)’
’draw ylab Sigma’
*
* std. dev. / mean density
’set vpage 4.0 8.0 4.5 8.5’
’set gxout line’
’set ylevs 1 0.5 0.2 0.1 0.05 0.02 0.01 0.005 0.002 0.001 0.0005 0.0002 0.0001’
’set cmark 0’
’set line 1’
’set grads off’
’display ’nsds’*sdrho.2(t=1)/rho’
’draw xlab ’nsds’Std.Dev./Mean Density’
’draw ylab Sigma’
*
* zonal wind
’set vpage 0.0 4.0 0.5 4.5’
’set gxout line’
’set ylevs 1 0.5 0.2 0.1 0.05 0.02 0.01 0.005 0.002 0.001 0.0005 0.0002 0.0001’
’set cmark 0’
’set line 1’
’set grads off’
’display u’
’set cmark 0’
’set cstyle 5’
’set ccolor 1’
’set grads off’
’display u + ’nsds’*sdu.2(t=1)’
’set cmark 0’
’set cstyle 5’
’set ccolor 1’
’set grads off’
’display u - ’nsds’*sdu.2(t=1)’
’draw xlab Zonal Wind (m/s)’

’draw ylab Sigma’
*
* meridional wind
’set vpage 4.0 8.0 0.5 4.5’
’set gxout line’
’set ylevs 1 0.5 0.2 0.1 0.05 0.02 0.01 0.005 0.002 0.001 0.0005 0.0002 0.0001’
’set cmark 0’
’set line 1’
’set grads off’
’display v’
’set cmark 0’
’set cstyle 5’
’set ccolor 1’
’set grads off’
’display v + ’nsds’*sdv.2(t=1)’
’set cmark 0’
’set cstyle 5’
’set ccolor 1’
’set grads off’
’display v - ’nsds’*sdv.2(t=1)’
’draw xlab Meridional Wind (m/s)’
’draw ylab Sigma’
*
* Finished plots, write and close metafile
’print’
’disable print’
return
*
*--------------------------------------------------
function max(a, b)
* return maximum of a and b
if (a >= b)
   return(a)
else
   return(b)
endif
*
*--------------------------------------------------
function min(a, b)
* return minimum of a and b
if (a <= b)
   return(a)
else
   return(b)
endif
*
*--------------------------------------------------
function nint(a)
* return the nearest integer to a
i = 0
if (a >= 0)
   a = a + 0.5
else
   a = a - 0.5
endif
while (i < 40)
   i = i + 1
   if (substr(a, i, 1) = ’.’)
      a = substr(a, 1, i-1)
      break
   endif
endwhile
return(a)
*
*--------------------------------------------------
function round(a, n)
* round a to n decimal places
’display pow(10, ’n’)’
scale = subwrd(result, 4)
a = nint(a*scale)/scale
return(a)

*
* section_lat.gs
*
* This script is similar to profile.gs, except that a range of latitudes
* can be given to produce a sigma-latitude section through the database.
*
* The longitude, first and last latitude and local solar time are supplied
* as arguments.  If the latitude range given is too small, it is expanded
* to include several database gridpoints; otherwise profile.gs could be
* used.  The longitude should be given in degrees east and the time in
* hours (00..24).  This script must be edited to select a dust scenario and
* season and to change the title accordingly.
*
* GrADS should be run in portrait mode.
*
function main(args)
*
* Set longitude (-180..180 degrees east), latitude (-90..90 degrees north)
* and local solar time (00..24 hours) for profiles
plon = subwrd(args,1)
if (plon > 180); plon = plon - 360; endif
if (plon < -180); plon = plon + 360; endif
plat1 = subwrd(args,2)
plat2 = subwrd(args,3)
if (plat2 - plat1 < 10)
   if (plat1 < 80)
      plat2 = plat1 + 10
   else
      plat1 = plat2 - 10
   endif
endif
ptime = subwrd(args,4)
if (ptime = ’’)
   say ’Four arguments are required for section_lat.gs’
   say ’run section_lat <longitude> <latitude1> <latitude2> <time>’
   say ’longitude = -180..180 (degrees east)’
   say ’latitude = -90..90 (degrees north), lat1 < lat2’
   say ’local solar time = 00..24 (hours)’
endif
*
* Choose data files for means and deviations according to dust scenario and
* season required and modify the main title accordingly
mefile = ’/home/ares/md1/esa/mcd/data/viks04me.ctl’
sdfile = ’/home/ares/md1/esa/mcd/data/viks04sd.ctl’
title1 = ’MCD Viking dust, Ls = 90-120.’
*
* Choose the number of standard deviations for the confidence limits
nsds = 1
*
* Add profile location and time to title
title2 = ’Section location:  lon = ’plon’E,  ’
title2 = title2%’lat = ’plat1’N to ’plat2’,  ’
title2 = title2%’local time = ’ptime’hrs.’
*
* If no data files are already open, open them,
* or if some are open check that they are what you expect them to be
’query files’
if (sublin(result, 1) = ’No files open’)
   ’open ’mefile
   ’open ’sdfile
else
   file1 = sublin(result, 2)
   file1 = subwrd(file1, 2)
   file2 = sublin(result, 5)
   file2 = subwrd(file2, 2)
   if ((file1!=mefile)|(file2!=sdfile))
      say ’The files open in grads are not consistent with profile.gs’
      say ’It is advisable to quit this session of grads and start again’
      return
   endif
endif
*
* Open metafile for writing output
’enable print sectlat_’plon’_’plat1’_’plat2’_’ptime’.gx’
*

* Convert local time into universal time at lon=0
ptime = ptime - plon/15.0
if (ptime >= 24.0)
   ptime = ptime - 24.0
else
   if (ptime < 0.0); ptime = ptime + 24.0; endif
endif
*
* Choose the closest grid and time points present in the database.
’set dfile 1’
’set lon ’plon
’set lat ’plat1’ ’plat2
ptime = (ptime + 1.0)/2.0
’set t ’ptime
*
* Title information
’set vpage 0.0 8.0 10.0 11.0’
’draw string 1.0 0.5 ’title1
’draw string 1.0 0.25 ’title2
*
* --- Surface data ---
’set z 1’
*
* surface pressure
’set vpage 0.0 4.0 5.25 10.0’
’set parea 1.5 8.0 1.0 9.59’
’set gxout line’
’set grads off’
’display ps’
’draw title Surface Pressure (Pa)’
’draw xlab Latitude’
’draw ylab Surface Pressure (Pa)’
*
* --- Vertical profiles ---
’set z 1 25’
’set zlog on’
*
* temperature
’set vpage 4.0 8.0 5.25 10.0’
’set gxout contour’
’set ylevs 1 0.5 0.2 0.1 0.05 0.02 0.01 0.005 0.002 0.001 0.0005 0.0002 0.0001’
’set grads off’
’display t’
’draw title Temperature (K)’
’draw xlab Latitude’
’draw ylab Sigma’
*
* zonal wind
’set vpage 0.0 4.0 0.5 5.25’
’set gxout contour’
’set ylevs 1 0.5 0.2 0.1 0.05 0.02 0.01 0.005 0.002 0.001 0.0005 0.0002 0.0001’
’set grads off’
’display u’
’draw title Zonal Wind (m/s)’
’draw xlab Latitude’
’draw ylab Sigma’
*
* meridional wind
’set vpage 4.0 8.0 0.5 5.25’
’set gxout contour’
’set ylevs 1 0.5 0.2 0.1 0.05 0.02 0.01 0.005 0.002 0.001 0.0005 0.0002 0.0001’
’set grads off’
’display v’
’draw title Meridional Wind (m/s)’
’draw xlab Latitude’
’draw ylab Sigma’
*
* Finished plots, write and close metafile
’print’
’disable print’
return

*
* zonal_sdtuv.gs
*
* Plots the zonal mean standard deviation of temperature, zonal and
* meridional velocity for one season.  GrADS should be run in portrait mode
* for this script.
*
’open /home/ares/md1/esa/mcd/data/viks04sd’
’set t 1’
’set lat -90 90’
’set z 1 25’
’set x 1’
’set zlog on’
’set vpage 0 8.5 7 11’
’set parea 1 7  1 3.5’
’set grads off’
’set ylevs 1 0.5 0.2 0.1 0.05 0.02 0.01 0.005 0.002 0.001 0.0005 0.0002 0.0001’
’define Tave = ave(sdt, lon=-180, lon=180, -b)’
’set gxout shaded’
’set cint 1’
’display Tave’
’set gxout contour’
’set ccolor 0’
’set cint 1’
’display Tave’
’draw xlab Zonal Mean Std. Dev. Temperature (K)’
’draw ylab Sigma’
’draw title MCD Viking dust, Ls = 90-120.’
’set vpage 0 8.5 3.5 7.5’
’set grads off’
’set ylevs 1 0.5 0.2 0.1 0.05 0.02 0.01 0.005 0.002 0.001 0.0005 0.0002 0.0001’
’define Uave = ave(sdu, lon=-180, lon=180, -b)’
’set gxout shaded’
’set cint 5’
’display Uave’
’set gxout contour’
’set ccolor 0’
’set cint 5’
’display Uave’
’draw xlab Zonal Mean Std. Dev. Zonal Wind U (m/s)’
’draw ylab Sigma’
’set vpage 0 8.5  0. 4’
’set grads off’
’set ylevs 1 0.5 0.2 0.1 0.05 0.02 0.01 0.005 0.002 0.001 0.0005 0.0002 0.0001’
’define Vave = ave(sdv, lon=-180, lon=180, -b)’
’set gxout shaded’
’set cint 5’
’display Vave’
’set gxout contour’
’set ccolor 0’
’set cint 5’
’display Vave’
’draw xlab Zonal Mean Std. Dev. Meridinal Wind V (m/s)’
’draw ylab Sigma’

*
* zonal_tuv.gs
*
* Plots the zonal mean and diurnal time mean temperature, zonal and
* meridional velocity for one season.  GrADS should be run in portrait mode
* for this script.
*
’open /home/ares/md1/esa/mcd/data/viks04me’
’set t 1’
’set lat -90 90’
’set z 1 25’
’set x 1’
’set zlog on’
’set vpage 0 8.5 7 11’
’set parea 1 7  1 3.5’
’set grads off’
’set ylevs 1 0.5 0.2 0.1 0.05 0.02 0.01 0.005 0.002 0.001 0.0005 0.0002 0.0001’
’define Tave = ave(ave(T, lon=-180, lon=180, -b), t=1, t=12)’
’set gxout shaded’
’set rbrange 130 250 ’
’set cint 10’
’display Tave’
’set gxout contour’
’set ccolor 0’
’set cint 10’
’display Tave’
’draw xlab Zonal/Diurnal Mean Temperature (K)’
’draw ylab Sigma’
’draw title MCD Viking dust, Ls = 90-120.’
’set vpage 0 8.5 3.5 7.5’
’set grads off’
’set ylevs 1 0.5 0.2 0.1 0.05 0.02 0.01 0.005 0.002 0.001 0.0005 0.0002 0.0001’
’define Uave = ave(ave(U, lon=-180, lon=180, -b), t=1, t=12) ’
’set gxout shaded’
’set rbrange -140 140’
’set cint 20’
’display Uave’
’set gxout contour’
’set ccolor 0’
’set cint 20’
’display Uave’
’draw xlab Zonal/Diurnal Mean Zonal Wind (m/s)’
’draw ylab Sigma’
’set vpage 0 8.5  0. 4’
’set grads off’
’set ylevs 1 0.5 0.2 0.1 0.05 0.02 0.01 0.005 0.002 0.001 0.0005 0.0002 0.0001’
’define Vave = ave(ave(V, lon=-180, lon=180, -b), t=1, t=12)’
’set gxout shaded’
’set rbrange -30 30’
’set cint 5’
’display Vave’
’set gxout contour’
’set ccolor 0’
’set cint 5’
’display Vave’
’draw xlab Zonal/Diurnal Mean Meridinal Wind (m/s)’
’draw ylab Sigma’